
Revit API 入門実習     

 

    Lab 7 – 共有パラメータ  

Created by A. Nagy, May 2011 

Updated by DevTech AEC WG 

Last modified: 6/9/2017 

<VB.NET>VB.NETバージョン</VB.NET>  

目的:この実習では、共有パラメータを作成する方法を学習します。学習する項目は次のとおりです。 

         共有パラメータ グループを作成する 

         共有パラメータ定義を作成する 

         それを特定のカテゴリにバインドする 例 > ドア 

この実習の実装と確認の手順は、下記のとおりです: 

1.    新しい外部コマンドを定義する 

2.   ヘルパー関数を作成する 

3.    ドア用の共有パラメータを作成する 

4.    ドキュメント毎のパラメータを作成する 

5.    サマリ 

  

1.   新しい外部コマンドを定義する 

現在のプロジェクトに新しい外部コマンドを追加します。 

1.1  新しいファイルを追加して、プロジェクトに新しい外部コマンドを定義します。ファイル名とクラス名は、下記の

ようにしてください: 

         ファイル名:  7_SharedParameter.vb 

         コマンド クラス名:SharedParameter  

  

 



要求される名前空間: 

いままで使用した名前空間に加えて、次の名前空間を加えてください: 

         System.IO 

  

さらに、作成するパラメータのプロパティを保持するいくつかの定数を宣言します。 

  
<VB.NET> 
  Const kSharedParamsGroupAPI As String = "API Parameters" 

  Const kSharedParamsDefFireRating As String = "API 耐火等級" 
  Const kSharedParamsPath As String = "C:\temp\SharedParams.txt" 
  
  Public Function Execute(ByVal commandData As ExternalCommandData, ByRef 

message As String, ByVal elements As ElementSet) _ 
    As Result _ 
    Implements IExternalCommand.Execute 
    Dim uidoc As UIDocument = commandData.Application.ActiveUIDocument 
    Dim app As Application = commandData.Application.Application 
    Dim doc As Document = uidoc.Document 
 </VB.NET> 

  

2.   ヘルパー関数を作成する 

最初に、共有パラメータ ファイルを返す関数を定義する必要があります。 この関数は共有パラメータ ファイルが

存在しない際に、新しい共有パラメータ ファイルを作成する機能を持つものとします。 

  
<VB.NET>  
  Public Shared Function GetSharedParamsFile(ByVal app As Application) _ 
    As DefinitionFile 
  
    ' Get current shared params file name 
  
    Dim sharedParamsFileName As String 
    Try 
      sharedParamsFileName = app.SharedParametersFilename 
    Catch ex As Exception 
      TaskDialog.Show("Get shared params file", _ 
                      "No shared params file set:" + ex.Message) 
      Return Nothing 
    End Try 
  
    If 0 = sharedParamsFileName.Length Or _  
      Not System.IO.File.Exists(sharedParamsFileName) Then 
  
      Dim stream As StreamWriter 
      stream = New StreamWriter(kSharedParamsPath) 
      stream.Close() 
      app.SharedParametersFilename = kSharedParamsPath 



      sharedParamsFileName = app.SharedParametersFilename 
    End If 
  
    ' Get the current file object and return it 
  
    Dim sharedParametersFile As DefinitionFile 
    Try 
      sharedParametersFile = app.OpenSharedParameterFile() 
    Catch ex As Exception 
      TaskDialog.Show("Get shared params file", _  
                      "Cannnot open shared params file:" + ex.Message) 
      sharedParametersFile = Nothing 
    End Try 
  
    Return sharedParametersFile 
  End Function 
 </VB.NET> 
  

次に、パラメータ グループが既に存在するかチェックして、それを返す関数を定義する必要があります。 もし、パ

ラメータ グループが存在しない場合には、新しくパラメータを作成して返すものとします。 

  
<VB.NET>  
Public Shared Function GetOrCreateSharedParamsGroup( _ 

    ByVal sharedParametersFile As DefinitionFile, ByVal groupName As String) 

_ 
    As DefinitionGroup 
  
    Dim g As DefinitionGroup = sharedParametersFile.Groups.Item(groupName) 
    If g Is Nothing Then 
      Try 
        g = sharedParametersFile.Groups.Create(groupName) 
      Catch generatedExceptionName As Exception 
        g = Nothing 
      End Try 
    End If 
  
    Return g 
  
  End Function 

 </VB.NET> 
  

最後に、与えられたパラメータ定義が既に存在するかチェックして、それを返す関数を定義します。 もし、パラメ

ータ定義が存在しない場合には、新しくパラメータを作成して返すものとします。 

 
<VB.NET> 
    Public Shared Function GetOrCreateSharedParamsDefinition( _ 

      ByVal defGroup As DefinitionGroup, _ 

      ByVal defType As ParameterType, _ 

      ByVal defName As String, _ 

      ByVal visible As Boolean) As Definition 

  

        Dim definition As Definition = defGroup.Definitions.Item(defName) 

        If definition Is Nothing Then 

            Try 

                ''Public Function Create(name As String, type As 



Autodesk.Revit.DB.ParameterType, visible As Boolean)  

        ''As Autodesk.Revit.DB.Definition' is obsolete:  

        'This method is deprecated in Revit 2015.  

        'Use Create(Autodesk.Revit.DB.ExternalDefinitonCreationOptions) 

instead' 

 

        'definition = defGroup.Definitions.Create(defName, defType, visible) 

 

        ' updated for Revit 2015 

        Dim extDefCrOptions As ExternalDefinitionCreationOptions _ 

          = New ExternalDefinitionCreationOptions(defName, defType) 

 

        extDefCrOptions.Visible = True 

        definition = defGroup.Definitions.Create(extDefCrOptions) 

  

            Catch generatedExceptionName As Exception 

                definition = Nothing 

            End Try 

        End If 

        Return definition 

    End Function 

</VB.NET> 
 
 

3.   ドア用の共有パラメータを作成する 

 

ドア用の共有パラメータを作成するために必要となるのは、共有パラメータ ファイル、パラメータ定義グループ、パ

ラメータ定義と、それをドア カテゴリに結び付ける作業です。 



  

最初に、共有パラメータ ファイルを取得して、パラメータ グループが存在することを確かめます。 

  
<VB.NET> 
    ' Get the current shared params definition file 
    Dim sharedParamsFile As DefinitionFile = GetSharedParamsFile(app) 
    If sharedParamsFile Is Nothing Then 
      message = "Error getting the shared params file." 
      Return Result.Failed 
    End If 
  
    ' Get or create the shared params group 
    Dim sharedParamsGroup As DefinitionGroup = _ 
      GetOrCreateSharedParamsGroup(sharedParamsFile, kSharedParamsGroupAPI) 
    If sharedParamsGroup Is Nothing Then 
      message = "Error getting the shared params group." 
      Return Result.Failed 
    End If 

 </VB.NET> 
  

次に、必要とする(ドア)カテゴリを取得して、共有パラメータ定義を作成します。 

  
<VB.NET> 
    Dim cat As Category = _ 
      doc.Settings.Categories.Item(BuiltInCategory.OST_Doors) 
  
    ' Visibility of the new parameter: 
    ' Category.AllowsBoundParameters property indicates if a category can  
    ' have shared or project parameters. If it is false, it may not be bound  
    ' to shared parameters using the BindingMap.  
    ' Please note that non-user-visible  
    ' parameters can still be bound to these categories.  
  
    Dim visible As Boolean = cat.AllowsBoundParameters 
  
    ' Get or create the shared params definition 
  
    Dim fireRatingParamDef As Definition = _ 
      GetOrCreateSharedParamsDefinition( _ 
        sharedParamsGroup, ParameterType.Number, _ 
        kSharedParamsDefFireRating, visible) 
  
    If fireRatingParamDef Is Nothing Then 
      message = "Error in creating shared parameter." 
      Return Result.Failed 
    End If 
 </VB.NET> 
  

共有パラメータ定義が作成できたら、それをドア カテゴリにバインドすることができます。 

  
<VB.NET> 
    ' Create the category set for binding and add the category 
    ' we are interested in, doors or walls or whatever: 
  
    Dim catSet As CategorySet = app.Create.NewCategorySet() 



    Try 
      catSet.Insert(cat) 
    Catch generatedExceptionName As Exception 
      message = String.Format( _  
        "Error adding '{0}' category to parameters binding set.", cat.Name) 
      Return Result.Failed 
    End Try 
  
    ' Bind the param 
  
    Try 
      Dim binding As Binding = app.Create.NewInstanceBinding(catSet) 
  
      ' We could check if already bound, but looks like Insert will  
      ' just ignore it in such case 
  
      doc.ParameterBindings.Insert(fireRatingParamDef, binding) 
    Catch ex As Exception 
      message = ex.Message 
      Return Result.Failed 
    End Try 
 </VB.NET> 
  

さて、アドインを実行して確認してみましょう。ユーザ インターフェースでドアを選択すると、最初に示したプロパティ

パレットのイメージと同じように、追加したパラメータがプロパティ パレットに表示されることを確認できるはずです。 

 

もしアドインの実行時にエラーが発生する場合は、編集しようとしている共有パラメータファイルが読み取り専用と

なっている可能性があります。その場合は、GetSharedParamsFile メソッドの、sharedParamsFileName = 

app.SharedParametersFilename を囲む ｔｒｙ – catch をコメントアウトして、再度実行してみてください。 

  



4.   ドキュメント毎にパラメータを作成する 

   

 

ドキュメント毎のパラメータに表示/非表示を作成する新しいコマンドを追加します。ドキュメント毎のパラメータが

意味するのは、プロジェクト内に単一のインスタンスを持つ要素に追加されるパラメータ、すなわち、プロジェクト

情報の要素のことです。前の実習と比較して、唯一の違いは、今回、作成したパラメータ定義を

BuiltInCategory.OST_ProjectInformation カテゴリにバインドする点です。 

さらに、FilteredElementCollector を使用してプロジェクト内でプロジェクト情報要素を検索、もしくは直接

Document.ProjectInformation プロパティから要素を取得し、新しく作成したパラメータ値を設定してください。 

5.   サマリ 

この実習では、共有パラメータの作成方法を学習しました。学習した内容は次のとおりです。 

         共有パラメータ グループを作成する 

         共有パラメータ定義を作成する 

         それを特定のカテゴリにバインドする 例 > ドア 

